

ให้เป็นลอจิก 1 เพราะดูตามรูปจะเห็นว่าบิต ACIE นี้ต่อเข้าโดยตรงกับขาอินพุตของแอนด์เกต เอาต์พุตอีกของส่วน Interrupt select นี้จะไปปรากฏในบิต ACI เพื่อบอกว่าขณะนี้เกิดการอินเทอร์รัพท์ขึ้น(บิต I ใน SREG ต้องเป็น 1 และ ACIE ต้องเป็น 1) ต่อไปจะให้ดูรายละเอียดของบิตต่างๆ ที่กล่าวมาข้างต้น ในรีจิสเตอร์ ACSR

```

+-----+
| ACSR | - Analog comparator Control and Status Register
+-----+

```

Bit	7	6	5	4	3	2	1	0
\$28	ACD	-	ACO	ACI	ACIE	ACIC	ACIS1	ACIS0
read/write	r/w	r	r	r/w	r/w	r/w	r/w	r/w
initial value	0	0	0	0	0	0	0	0


- <> Bit 7 - ACD : Analog Comparator Disable
บิตนี้ถ้าเป็น 0 จะทำให้สวิตช์(ด้านบนตัวออปแอมป์) ต่อกับ (ON) ทำให้มีไฟจาก Vcc มาเลี้ยงออปแอมป์ ถ้าเราไม่ใช้ก็ควรให้บิตนี้เป็น 1 เพื่อจะได้ให้สวิตช์ตัด(OFF) ทำให้ประหยัดกำลังงาน ในสภาวะปกติบิตนี้จะเป็นลอจิก 0
- <> Bit 6 - reserved bit
- <> Bit 5 - ACO : Analog Comparator Output
เป็นเอาต์พุตของวงจรเปรียบเทียบแรงดัน ถ้าขาอินพุตบวกของออปแอมป์มีแรงดันสูงกว่าขาอินพุตลบ บิต ACO จะเป็น 1 ถ้าในทางกลับกันบิต ACO จะเป็นลอจิก 0
- <> Bit 4 - ACI : Analog Comparator Interrupt flag
บิตนี้จะกลายเป็น 1 เมื่อเริ่มเกิดการอินเทอร์รัพท์จากอนาลอกคอมพาราเตอร์ ละจะเป็น 0 โดยฮาร์ดแวร์ภายใน เมื่อเกิดการทำงานของโปรแกรมย่อยที่รองรับการอินเทอร์รัพท์
- <> Bit 3 - ACIE : Analog Comparator Interrupt Enable
ถ้าบิตนี้เป็น 1 และบิต I ใน SREG เป็นลอจิก 1 จะเป็นการยอมให้เกิดการอินเทอร์รัพท์ของอนาลอกคอมพาราเตอร์
- <> Bit 2 - ACIC : Analog Comparator Input Capture Enable
เมื่อถูกเซตให้เป็น 1 จะทำให้อนาลอกคอมพาราเตอร์ต่อไปเพื่อทริก Timer/Counter 1 (บิต TICIE1 ใน TIMSK ต้องเป็น 1 ด้วย)
- <> Bit 1,0 : Analog Comparator Interrupt Mode Select
บิตทั้งสองนี้จะกำหนดว่า การเปลี่ยนแปลงเอาต์พุตของอนาลอกคอมพาราเตอร์แบบใดที่จะทำให้เกิดการอินเทอร์รัพท์

ACIS1	ACIS0	Interrupt mode
0	0	เกิดการอินเทอร์รัพท์ทุกครั้งที่ output toggle
0	1	สงวนไว้
1	0	เกิดการอินเทอร์รัพท์ที่ขอบขาลง
1	1	เกิดการอินเทอร์รัพท์ที่ขอบขาขึ้น

(2) เริ่มต้นกับ Analog Comparator

เราจะทดลองใช้งาน Analog Comparator เบื้องต้น ให้เป็นวงจรเปรียบเทียบแรงดันง่ายๆ ว่าอินพุตใดของออปแอมป์มีแรงดันสูงกว่ากัน โดยจะยังไม่ใช้การอินเทอร์รัพท์

ประกอบวงจรดังรูป(ไม่ได้แสดงวงจร ISP และ crystal)



เราต่อ R1,R2 เป็นวงจรแบ่งแรงดัน(Voltage Divider) ได้แรงดันครึ่งหนึ่งของแหล่งจ่ายไฟคือ 2.5 โวลต์ ต่ วงจร,จ่ายไฟ แล้วลองใช้โวลต์มิเตอร์วัดที่ขา 12 จะได้ 2.5 โวลต์ ถ้าได้ต่างไปจากนี้มาก ให้ตรวจสอบการต่ วงจรใหม่ สำหรับ VR1 ถูกต่อเป็นวงจรเปรียบเทียบแรงดันเช่นเดียวกันแต่เป็นวงจรแบ่งแรงดันที่สามารถปรับค่าได้ตั้งแต่ 0 จนถึง +5V ให้ใช้โวลต์มิเตอร์วัดที่ขา 13 หรือที่เขากลางของตัวต้านทานเก็อกมา ใช้ไขควงหมุนปรับ VR1 ไปมา แรงดันจะเปลี่ยนแปลงไปตามการหมุนไขควง คือปรับได้ 0 ถึง +5V ถ้าไม่ได้ เช่น ลดแรงดันลงไปได้ไม่ถึง 0 โวลต์ อาจะเกิดจากการชำรุดตัวต้านทานปรับค่าได้

เราต่อ LED เข้ากับขา PD5 จะให้ติดเมื่อแรงดันขาอินพุตบวกมากกว่าขาอินพุตลบ และดับเมื่ออินพุตลบมากกว่าอินพุตบวก การต่อขาที่ตรวจสอบให้ตีว่าขาใดเป็นขาเอาต์พุตขาใดเป็นขา ดาโธด(LED แบบความสว่างสูง มักจะมีขาที่กลับกับแบบธรรมดา) ใช้โหม้มิเตอร์วัดดูจะดีที่สุด เป็นการตรวจสอบไปในตัวด้วยว่า LED ยังติดอยู่หรือปล่าว

(3) โปรแกรม(1)

ต่อไปเราจะเขียนโปรแกรมเพื่อให้วงจรทำงานตามที่ได้กำหนดไว้ ใช้วิธีตรวจสอบบิต ACO ของรีจิสเตอร์ ACSR ซึ่งเป็นเอาต์พุตของ Analog Comparator โดยตรง LED1 จะสว่าง เมื่อแรงดันขา 12 มากกว่าแรงดันอ้างอิงที่ขา 13

```

=====
===== ==
V(pin13) 2.5V ----- Vref
===== =
==
V(pin12) 0V =
=====
+-----+ LED1="ON"
| |
VLED1 0V -----+ LED1="OFF"

;*****
; TITLE : _ANALOG1.ASM
; Analog Comparator(1 - ACO sense methode)
; XTAL = 4.0MHz
; MCU = AT90S2313,Atmel corperation,www.atmel.com
; copyright (c) 2001 by Gumtorn Ruanfaigad.Allright reserved.
; Electronics devision.RIT Chiangmai campus.tel 053-221576
;*****

.include "2313def.inc" ;
.DEVICE AT90S2313 ;
.DSEG ;
.org 100 ;
L_data: .byte 1 ;
;
.CSEG ;
.ORG 0x0000 ;
; =====
; interrupt vector table
; =====

rjmp F_main ; #0 RESET
reti ; #1 INTO
reti ; #2 INT1

```

```

reti                ; #3 TIMER1 CAPT11
reti                ; #4 TIMER1 COMP1
reti                ; #5 TIMER1 OVF1
reti                ; #6 TIMER0 OVFO
reti                ; #7 UART,RX
reti                ; #8 UART,UDRE
reti                ; #9 UART,TX
reti                ; #10 ANA_COMP


;=====
;void main(void)
;=====
F_Main:             ; main function start here
    ldi    r16,    low(RAMEND)    ; set stack pointer
    out    SPL,    r16            ; -
    sbi    DDRD,   PD5           ; PD5=output for driving LED
L_Main_Loop:       ;
    sbic   ACSR,   ACO           ;
    sbi    PORTD,  PD5           ;
    sbis   ACSR,   ACO           ;
    cbi    PORTD,  PD5           ;
    rjmp   L_MAIN_LOOP          ;
;.....

```

เมื่ออัปเดตโปรแกรมลง MCU แล้ว ให้ทดลองปรับ VR1 ดู จะพบว่าถ้าแรงดันที่ขา 12 สูงกว่าขา 13 (2.5 โวลต์) จะทำให้ LED1 สว่าง แต่ถ้าแรงดันขา 12 ต่ำกว่าขา 13 LED1 จะดับ

(4) โปรแกรม(2)

ต่อไปเราจะเขียนโปรแกรมให้ทำงานโดย โดยใช้วิธีการอินเทอร์รัพท์(หมายเลข #10) ให้ทำงานในโหมดการทริกแบบ toggle(ACIS0=0,ACIS1=0) ถ้าปรับ VR1 ให้แรงดันขา 13 ผ่านระดับแรงดันอ้างอิง LED1 จะกระพริบครึ่งหนึ่ง และเมื่อปรับ VR1 ให้ย้อนกลับ มาผ่านจุดนี้อีก LED1 ก็จะกระพริบอีกครึ่งหนึ่งดังรูป



```

;*****
; TITLE : _ANALOG2.ASM
;       Analog Comparator(1 - Interrupt sense methode)
; XTAL  = 4.0MHz
; MCU   = AT90S2313,Atmel corperation,www.atmel.com
; copyright (c) 2001 by Gumtorn Ruanfaigad.Allright reserved.
; Electronics devision.RIT Chiangmai campus.tel 053-221576
;*****

.include      "2313def.inc"          ;
.DEVICE       AT90S2313              ;
.DSEG        ;
.org         100                     ;
L_data: .byte 1                      ;
;
.CSEG        ;
.ORG         0x0000                  ;
; =====
; interrupt vector table
; =====
        rjmp   F_main                ; #0 RESET
        reti   ;                     ; #1 INTO
        reti   ;                     ; #2 INT1
        reti   ;                     ; #3 TIMER1 CAPT11
        reti   ;                     ; #4 TIMER1 COMP1
        reti   ;                     ; #5 TIMER1 OVF1
        reti   ;                     ; #6 TIMER0 OVFO
        reti   ;                     ; #7 UART,RX
        reti   ;                     ; #8 UART,UDRE
        reti   ;                     ; #9 UART,TX
        rjmp   F_ACMP                 ; #10 ANA_COMP

;=====
;Function : Analog Comparator Interrupt Routine
;{rin} none
;{rout} none
;{port} output=PD5
;=====
F_ACMP:                                     ;
        sbi    PORTD, PD5              ; LED1="ON"
        reti   ;
;.....

```

```

;=====
;void main(void)
;=====
F_Main:                                ; main function start here
    ldi    r16,    low(RAMEND)         ; set stack pointer
    out    SPL,    r16                 ; -
    sbi    DDRD,   PD5                 ; PD5=output for driving LED
    cbi    ACSR,   ACIE                 ; Analog Comparator Interrupt Disable
    cbi    ACSR,   ACIS0                ; ACMP. Interrupt sense = toggle
    cbi    ACSR,   ACIS1                ; -
    sbi    ACSR,   ACIE                 ; Analog Comparator Interrupt Enable
    sei                                ; Global Interrupt Enable
L_MainLoop:                             ;
    cbi    PORTD,  PD5                 ; LED1="OFF"
    rjmp   L_MainLoop                 ; wait for interrupt
;.....

```

ทำไม LED1 จึงกระพริบ?

=====

ในหลูปของ L_MainLoop เราสั่งให้ PD5 เป็น 0 ตลอดเวลา แต่ในอินเตอร์รัพท์รูทีน เราสั่งให้ PD5 เท่ากับ 1 ดังนั้น LED จึงกระพริบซุกซุก ครั้งที่มีการอินเตอร์รัพท์เข้ามา